APPENDICE 2
LA SOLUZIONE DI UN PROBLEMA DI GEOMETRIA ANALITICA

Nella presente appendice viene illustrato un metodo di implementazione in Qbasic dei processi di calcolo per la soluzione di un problema caratteristico di geometria analitica.

A2.1 Sui problemi di geometria analitica

La tematica relativa ai problemi di geometria analitica è molto vasta. Innumerevoli sono gli esercizi che si possono proporre, ciascuno con un particolare grado di difficoltà. Non è possibile pertanto spaziare su questo argomento in una ridotta appendice del testo.
Ci limiteremo perciò a prendere in esame la soluzione di un esercizio di media complessità nell'intento di mostrare al lettore le procedure generali per arrivare all'implementazione in Qbasic degli algoritmi che sviluppano il calcolo numerico e la presentazione grafica che risolve il problema proposto.

A2.2 Proposizione del problema

Si determinino le equazioni delle rette passanti per un punto Po e tangenti ad una circonferenza C avente il centro nell'origine degli assi.
Si calcolino le coordinate dei punti di tangenza individuando l'equazione della retta passante per detti punti.
Si tracci infine il grafico risolutivo in un sistema di assi cartesiani calibrato.

A2.3 Soluzione analitica del problema

La soluzione analitica del problema prevede inizialmente l'impostazione dell'equazione generale di una retta passante per un punto Po di coordinate (xo ; yo); l'equazione della retta è:

$$(y - yo) / (x - xo) = m$$

da cui

$$y = m\,x - m\,xo + yo$$

dove **m** è il coefficiente angolare
(- m xo + yo) = **n** è il parametro

La retta così specificata deve essere messa a sistema con l'equazione della circonferenza in cui si cercano i punti di tangenza:

$$\begin{cases} y = m\,x - m\,xo + yo \\ y^2 + x^2 = R^2 \end{cases}$$

Si risolve il sistema in x ponendo il secondo membro della prima equazione al posto della variabile dipendente y dell'equazione della circonferenza; si ha:

$$(m\,x - m\,xo + yo)^2 + x^2 = R^2$$

sviluppando si ottiene l'equazione di secondo grado in x:

$$x^2 (m^2 + 1) - 2 m x (m xo - yo) - R^2 + (m xo - yo)^2 = 0$$

Sappiamo che per avere la condizione di tangenza tra le rette e la circonferenza, l'equazione trovata deve avere radici reali e coincidenti; cioè deve essere nullo il discriminante D dell'equazione:

$$D = b^2 - 4 a c = 0$$

dove

$$a = (m^2 + 1)$$
$$b = - 2 m (m xo - yo)$$
$$c = - R^2 + (m xo - yo)^2$$

perciò si deve scrivere:

$$[- 2 m (m xo - yo)]^2 - 4 (m^2 + 1) [-R^2 + (m xo - yo)^2] = 0$$

la nuova equazione in m, sviluppata e risolta dà:

$$m = \left\{ - (2 xo yo) \pm [(- 2 xo yo)^2 - 4 (R^2 - xo^2) (R^2 - yo^2)]^{1/2} \right\} / 2 (R^2 - xo^2)$$

Come era previsto l'equazione fornisce due valori di m: m1; m2, sono infatti due le rette tangenti alla circonferenza data; i valori di m soddisferanno il problema soltanto se il punto Po sarà esterno alla circonferenza. Ciò è ovvio, altrimenti le rette sarebbero secanti e non sussisterebbe la condizione imposta D = 0.
Deve pertanto essere sempre verificata una delle seguenti disuguaglianze:

$$|xo| > R$$

$$|yo| \geq R$$

Il problema non ha soluzione numerica se $|xo| = R$ dato che si annulla il denominatore dell'espressione in m creando una condizione di infinito. In questa situazione il calcolo algebrico di m non è possibile; è il caso di una retta tangente alla circonferenza e parallela all'asse y.
L'espressione in m alla quale siamo pervenuti dovrà essere implementata in Qbasic per il calcolo dei valori di m1 ed m2 che interessano il nostro problema.
Il tracciamento delle rette tangenti alla circonferenza presuppone anche il calcolo dei valori di n1 ed n2 secondo l'equazione

$$n = yo - xo\, m$$

anche questa espressione dovrà essere implementata nella routine di programma.
Il problema si completerà con la determinazione delle coordinate (x1 ; y1) e (x2 ; y2) relative ai due punti, P1 e P2, di tangenza sulla circonferenza e con la stesura dell'equazione della retta passante per questi.
Le coordinate dei due punti sono le soluzioni del sistema iniziale dopo che in esso sono stati inseriti i valori calcolati di (m1 ; n1) e (m2 ; n2):

$$\begin{cases} y = m1\, x + n1 \\ y^2 + x^2 = R^2 \end{cases} \quad \text{per le coordinate di P1}$$

$$\begin{cases} y = m2\,x + n2 \\ y^2 + x^2 = R^2 \end{cases} \qquad \text{per le coordinate di P2}$$

Essendo D = 0 i due sistemi hanno le seguenti soluzioni:

Per P1 si ha $x1 = -\,m1\;n1 / (m1^2 + 1)$ $\qquad y1 = n1 / (m1^2 + 1)$

Per P2 si ha $x2 = -\,m2\;n2 / (m2^2 + 1)$ $\qquad y2 = n2 / (m2^2 + 1)$

La retta individuata da P1 e P2 ha equazione:

$$(y - y1) / (x - x1) = (y2 - y1) / (x2 - x1)$$

che sviluppata dà:

$$y = [(y2 - y1) / (x2 - x1)]\,x - x1\,[(y2 - y1) / (x2 - x1)] + y1$$

in cui $\quad m = m3 = (y2 - y1) / (x2 - x1)$

$\quad n = n3 = -\,x1\,[(y2 - y1) / (x2 - x1)] + y1$

A2.4 La compilazione del programma di calcolo

A questo punto non resta che trasformare in simbologia Qbasic ciò che abbiamo esplicitato in precedenza.
Per semplificare il lavoro di compilazione ci serviremo del programma dell'esercizio di paragrafo 3.28.1 relativo al tracciamento della circonferenza.
Aggiungeremo ad esso le istruzioni per l'inserimento delle coordinate di Po, del valore di R, dell'intervallo di variabilità della x, le istruzioni per il calcolo di (m1;n1), (m2;n2), (m3;n3), (x1;y1), (x2;y2), le istruzioni per il tracciamento delle due tangenti, le istruzioni per il tracciamento della retta passante per P1 e P2, nonché i comandi per la presentazione dei dati calcolati e per la presentazione grafica delle tre rette.

Le funzioni di calcolo sono computabili come segue:

per m1 e m2

```
m1 = (-(2*xo*yo) + SQR((2*xo*yo)^2 - 4*(R^2-xo^2)*(R^2-yo^2))/(2*(R^2-xo^2))

m2 = (-(2*xo*yo) - SQR((2*xo*yo)^2 - 4*(R^2-xo^2)*(R^2-yo^2))/(2*(R^2-xo^2))
```

per n1 e n2

```
n1 = yo-xo*m1

n2 = yo-xo*m2
```

per le coordinate dei punti di tangenza P1 e P2

x1 = - m1*n1/(m1^2+1)
y1 = n1/(m1^2+1)

x2 = - m2*n2/(m2^2+1)
y2 = n2/(m2^2+1)

le equazioni delle rette tangenti

yt1 = m1*x+n1

yt2 = m2*x+n2

l'equazione della retta passante per P1 e P2

m3 = (y2-y1)/(x2-x1)
n3 = (-x1*(y2-y1)/(x2-x1))+y1
y3 = m3*x+n3

l'equazione della circonferenza

C1 = SQR(R^2-x^2)
C2 = -SQR(R^2-x^2)

Con le corrispondenze simboliche sopra scritte compiliamo infine il programma di calcolo con la presentazione dei risultati numerici a sole due cifre decimali, più che sufficienti in questo tipo di esercizio:

```
SCREEN 9 ' impostazione del sistema grafico a 4 quadranti

FOR x = 0 TO 460 STEP 23

FOR y = 0 TO 320 STEP 2

PSET (x, y), 7

NEXT y

NEXT x

FOR y = 0 TO 320 STEP 16

FOR x = 0 TO 460 STEP 3

PSET (x, y), 7

NEXT x

NEXT y

LINE (0, 160)-(460, 160)

LINE (230, 0)-(230, 320)

FOR x = -10 TO 10 STEP .01                              &
```

```
NEXT x

LOCATE 1, 60: INPUT "xo="; xo ' richiesta introduzione ascissa di Po

LOCATE 2, 60: INPUT "yo="; yo ' richiesta introduzione ordinata di Po

LOCATE 3, 60: INPUT "R="; R ' richiesta introduzione raggio della circonferenza

LOCATE 4, 60: INPUT "Est="; e ' richiesta introduzione estensione intervallo assi cartesiani

LOCATE 5, 60: INPUT "step"; s ' richiesta introduzione incremento di calcolo

FOR x = -R TO R STEP s  ' impostazione del campo di variabilità della x per la circonferenza

C1 = SQR((R ^ 2) - (x ^ 2)) ' equazione della circonferenza (ramo superiore)

C2 = -SQR((R ^ 2) - (x ^ 2)) ' equazione della circonferenza (ramo inferiore)

PSET (230 + (230 / e) * x, 160 - (160 / e) * C1), 10 ' tracciamento ramo superiore di C in verde

PSET (230 + (230 / e) * x, 160 - (160 / e) * C2), 10 ' tracciamento ramo inferiore di C in verde

NEXT x ' rimanda all'istruzione FOR x = -R ...

' computazioni dei coefficienti angolari delle rette tangenti

m1 = (-(2 * xo * yo) + SQR((2 * xo * yo) ^ 2 - 4 * (R ^ 2 - xo ^ 2) * (R ^ 2 - yo ^ 2))) / (2 * (R ^ 2 - xo ^ 2))

m2 = (-(2 * xo * yo) - SQR((2 * xo * yo) ^ 2 - 4 * (R ^ 2 - xo ^ 2) * (R ^ 2 - yo ^ 2))) / (2 * (R ^ 2 - xo ^ 2))

n1 = yo - xo * m1 ' computazione dei parametri delle rette tangenti

n2 = yo - xo * m2

x1 = (-m1 * n1) / (m1 ^ 2 + 1) ' computazione coordinate punti di tangenza

x2 = (-m2 * n2) / (m2 ^ 2 + 1)

y1 = n1 / (m1 ^ 2 + 1)

y2 = n2 / (m2 ^ 2 + 1)

m3 = (y2 - y1) / (x2 - x1)  ' computazione coefficiente angolare retta passante per P1 e P2

n3 = -(x1 * (y2 - y1) / (x2 - x1)) + y1 ' computazione parametro retta passante per P1 e P2

FOR x = -e TO e STEP s ' impostazione del campo di variabilità della x per le tre rette

yt1 = m1 * x + n1 ' equazione prima retta tangente

yt2 = m2 * x + n2 ' equazione seconda retta tangente

y3 = m3 * x + n3 ' equazione retta passante per P1 e P2

PSET (230 + (230 / e) * x, 160 - (160 / e) * yt1), 14 ' tracciamento prima retta tangente in giallo

PSET (230 + (230 / e) * x, 160 - (160 / e) * yt2), 14 ' tracciamento seconda retta tangente in giallo

PSET (230 + (230 / e) * x, 160 - (160 / e) * y3), 3 ' tracciamento  retta passante per P1 e P2 in turchese        &
```

```
NEXT x ' rimanda all'istruzione For x= -e....

LOCATE 10, 60: PRINT "m1="; USING "###.##"; m1 ' presentazione dati calcolati
                                               ' a 3 interi e 2 decimali
LOCATE 11, 60: PRINT "n1="; USING "###.##"; n1

LOCATE 12, 60: PRINT "m2="; USING "###.##"; m2

LOCATE 13, 60: PRINT "n2="; USING "###.##"; n2

LOCATE 14, 60: PRINT "x1="; USING "###.##"; x1

LOCATE 15, 60: PRINT "y1="; USING "###.##"; y1

LOCATE 16, 60: PRINT "x2="; USING "###.##"; x2

LOCATE 17, 60: PRINT "y2="; USING "###.##"; y2

LOCATE 18, 60: PRINT "m3="; USING "###.##"; m3

LOCATE 19, 60: PRINT "n3="; USING "###.##"; n3
```

Utilizziamo il programma per la ricerca delle due rette passanti per Po, di coordinate xo = 4; yo = 7, tangenti alla circonferenza C di raggio R = 5.
Essendo verificata la disuguaglianza |yo| > R il problema è possibile.
Imponiamo al grafico una estensione (Est) degli assi cartesiani tale da rendere la visualizzazione completa di C e Po:
deve essere (Est) > |xo|; |yo|; R, si assume, per calibrare anche le scale, (Est) = 10 con s = .02 pari a 1000 punti di calcolo.
Premendo F5 si ha la richiesta di introduzione dati; dopo l'ultimo valore inserito si ottiene la presentazione di tutti gli elementi calcolati e la soluzione grafica del nostro problema così come mostrato in figura 78.

La schermata del P.C., riportata in figura, contiene quanto richiesto nella proposizione del problema:
-i valori di m1; n1 e m2; n2 per la costruzione analitica delle equazioni delle due rette tangenti la circonferenza
-le coordinate dei due punti di tangenza tra le rette e la circonferenza (x1 ; y1), (x2 ; y2)
-i valori di m3; n3 per la costruzione analitica della equazione della retta passante per P1 e P2
-il grafico risolutivo del problema tracciato in un sistema di assi cartesiani aventi scale calibrate, con intervalli in x e y ciascuno suddiviso in 10 unità

Le equazioni delle rette sono pertanto:

le tangenti

$$y = .4\,x + 5.39$$

$$y = -\,6.62\,x + 33.5$$

la retta passante per P1 e P2

$$y = .57\,x + 3.57$$

Per completare l'esercizio si deve osservare che se i punti P1 e P2 avessero ascisse uguali, x1 = x2, le espressioni per il calcolo di m3 ed n3 non sarebbero algebricamente computabili perché avrebbero i denominatori nulli.



Figura 78
Schermata risolutiva del problema dato

A2.5 Note

L'esercizio che abbiamo svolto, essendo a parametri variabili, propone l'impiego del programma in tutti i casi simili avendo come gradi di libertà le coordinate (xo ; yo) di Po e il raggio R della circonferenza.
Si consiglia di ripetere la computazione per diversi valori dei parametri per poter comprendere quale è l'efficacia del programma sviluppato.
Naturalmente per la soluzione di un caso più generale del problema, relativo al calcolo delle tangenti ad una circonferenza ovunque collocabile nel piano, è necessaria una più complicata impostazione analitica, seguendo però le tracce del lavoro già svolto; lavoro che può servire inoltre come guida per la soluzione di problemi di geometria analitica riguardanti curve diverse da quelle ora trattate.

BIBLIOGRAFIA

-HANDBOOK OF MATHEMATICAL FUNCTIONS
Abramowitz and Stegun Dover Publications, Inc. New York

-MANUALE MS-DOS Q BASIC
Microsoft Corporation

-PROGRAMMARE IN QUICKBASIC Don Inmann, Bob Albrecht
Mc Graw-HiII Italia Milano

-METODI MATEMATICI NELL'INGEGNERIA T.v.Karman, MA. Biot
Einaudi Torino

-ESERCIZI DI ANALISI MATEMATICA vol.I B.Finzi F.Morra
Libreria Editrice Politecnica Cesare Tamburini Milano

-ANALISI MATEMATICA volI. G.Moretti
Hoepli Milano

-ANALISI MATEMATICA volII parte prima G.Moretti
Hoepli Milano

-ANALISI MATEMATICA volII parte seconda G .Moretti
Hoepli Milano

-THE FOURIER INTEGRAL AND ITS APPLICATIONS A.Papoulis
Mc Graw-HiII New York

-What is the Fast Fourier Transform? W.T.Cochran, J.W. Coo]ey
IEEE Trans. Audio Electroacoust. vol. AU-15,pp.45-55, June 1967

-LA CORRELAZIONE C.DeI Turco La Moderna La Spezia

-IL PROBLEMA GEOMETRICO E LA GEOMETRIA ANALITICA R.Ferrauto
Soc. Editrice Dante Alighieri Città di Castello